DSGE モデル

はじめに

- DSGE モデルとは Dynamic Stochastic General Equilibrium Model の略で、日本語では「動学的確率的一般均衡モデル」と呼ばれます。経済学、特にマクロ経済学で政策分析と予測に使用される多変量時系列モデルです。
- モデルの形は連立方程式です。その構造は経済理論に基づくものであり、特に変数の将来的な値としての期待値がモデルの中で大きな役割を果たします。経済理論をそのまま用いてモデル構築を行うため、推定値は理論に沿った形で、ダイレクトに解釈できます。
- Stata には、非線形 DSGE モデルのための `dsgenl` コマンド、線形 DSGE モデルのための `dsge` コマンドが用意されています。
- この例題では、はじめに本来の非線形形式でモデルを記述し、`dsgenl` を用いてパラメータを推定します。次にモデルを線形変換し、`dsge` を用いてパラメータを推定します。一般的な DSGE モデルのフレームワークを理解することを目的とします。

✧ 本文中のコマンドをコピーし、Stata のコマンドウィンドウに貼り付けて実行できます。全ての操作のコマンドは、do ファイル dsge.do にまとめられています。

例題：非線形 DSGE モデル

非線形 DSGE モデルを書き換える

ここでは Woodford(2003, chap. 4)を参考にした次のモデルを利用して、DSGE モデルの利用方法を解説します。主に家計、企業、中央銀行の行動をモデル化します。これらの経済主体の行動を組みわせてインフレ率、産出量の成長率、金利をモデルで表現します。アカデミックな分野ではなじみのあるモデルだと思います。金融政策における政策決定のメカニズムを表現します。

- 現時点の生産量Y_tは、翌期の生産量の期待値Y_{t+1}とインフレ率$\prod_{t+1}$，そして現時点の名目利子率R_tによって決定するものとします。

$$\frac{1}{Y_t} = \beta E_t\left(\frac{1}{Y_{t+1}}\frac{R_t}{\Pi_{t+1}}\right) \tag{1}$$

- βは消費の遅延選考率を示すパラメータです。
- 企業の最適化行動によってインフレ率の定常状態からの偏差を、将来のインフレ率とその定常状態からの偏差の期待値に回帰する次の関係をもたらします。
- 右辺のY_tは生産量で、Z_tは生産量の自然水準です。

$$(\Pi_t - \Pi) + \frac{1}{\phi} = \phi\left(\frac{Y_t}{Z_t}\right) + \beta E_t(\Pi_{t+1} - \Pi) \qquad (2)$$

- ϕは企業の価格決定式にリンクするパラメータです。
- 企業の意思決定はインフレ率に影響されないものとします。
- 最後に中央銀行の政策決定式について考えます。
- 中央銀行はインフレ率や、ここでは利用しないいくつかの変数によって利子率を調整します。

$$\frac{R_t}{R} = \left(\frac{\Pi_t}{\Pi}\right)^{1/\beta} U_t \qquad (3)$$

- ここでRは利子率の定常状態の値であり、U_tはインフレ率以外のすべての要因とします。
- Woodford(2003)に従って$X_t = Y_t/Z_t$として(1)-(3)式を書き換えます。

$$1 = \beta E_t\left(\frac{X_t}{X_{t+1}}\frac{1}{G_t}\frac{R_t}{\Pi_{t+1}}\right) \qquad (4a)$$

$$(\Pi_t - \Pi) + \frac{1}{\phi} = \phi X_t + \beta E_t(\Pi_{t+1} - \Pi) \qquad (5a)$$

$$\frac{R_t}{R} = \left(\frac{\Pi_t}{\Pi}\right)^{1/\beta} U_t \qquad (6a)$$

- ここで$G_t = Z_{t+1}/Z_t$はZ_tの変化量を示す状態変数です。
- 最後に状態変数G_tとU_tの成長を表現する次の自己回帰位モデルを追加します。

$$\ln(G_{t+1}) = \rho_g \ln(G_t) + \xi_{t+1} \qquad (7a)$$

$$\ln(U_{t+1}) = \rho_u \ln(U_t) + \epsilon_{t+1} \qquad (8a)$$

- 変数ξ_{t+1}とϵ_{t+1}は状態変数のショックを示します。以上で非線形の DSGE モデルの定義は終了です。

データの用意

dsgenl コマンドと dsge コマンドを利用するためには、データが次の 3 つの条件を満たす必要があります。

1. tsset コマンドを利用して時間情報を設定する。
2. データの平均がゼロになるようにする。dsge または dsgenl を実行すると変数を定常状態からの階差に変換する。
3. 弱定常過程にある変数を利用する。

- サンプルデータ rates2.dta をインポートします。このデータには、価格水準と名目利子率などのマクロデータが入っています。Stata のコマンドウィンドウで次のコマンドを実行します。

```
use https://www.stata-press.com/data/r16/rates2, clear
```

- tsset コマンドで時間情報を設定します。変数 dateq が時間変数として自動認識されます。

```
tsset
```

結果画面に次のように表示されます。1947 年から 2017 年までの四半期データであることがわかります。

```
. tsset dateq
        time variable:  dateq, 1947q1 to 2017q1
                delta:  1 quarter
```

- データの内容を確認します。

```
describe
```

次のように表示されます。価格水準は gdpdef です。

```
Contains data from http://www.stata-press.com/data/r16/rates2.dta
  obs:           281                          Federal Reserve Economic Data - St. Louis Fed,
                                                2017-02-10
 vars:             5                          26 Apr 2019 21:22
─────────────────────────────────────────────────────────────────────────────────────────────
              storage   display    value
variable name   type    format     label      variable label
─────────────────────────────────────────────────────────────────────────────────────────────
datestr         str10   %-10s                 Observation date
daten           int     %td                   Numeric (daily) date
gdpdef          float   %9.0g                 GDP deflator GDPDEF
r               float   %9.0g                 Federal funds rate FEDFUNDS
dateq           int     %tq                   Quarterly date
─────────────────────────────────────────────────────────────────────────────────────────────
Sorted by: dateq
```

- 四半期データの場合、価格水準の対数階差を取って 400 倍し、インフレ率 p とします。L. はラグを意味する Stata の時系列演算子です。⇒PDF マニュアル[U] User's Guide - 11. Language syntax

  さらに、変数 p のラベルを"Inflation rate"とします。

```
generate p = 400*(ln(gdpdef) - ln(L.gdpdef))

label variable p "Inflation rate"
```

dsgenl コマンドでモデルを設定する

　非線形 DSGE モデルは(4a)-(8a)式によって定義されています。これをコマンドで示すと次のようになります。///は改行を示します。

```
. dsgenl (1 = {beta}*(x/F.x)*(1/g)*(r/F.p) )          ///
         (1/{phi} + (p-1) = {phi}*x + {beta}*(F.p-1))     ///
         ({beta}*r = p^(1/{beta})*u)                  ///
         (ln(F.u) = {rhou}*ln(u))                     ///
         (ln(F.g) = {rhog}*ln(g)),                    ///
         exostate(u g) observed(p r) unobserved(x)
```

- 各推定式をカッコで囲みます。
- 推定式に加えて、変数がモデル内で果たす役割を、オプションで設定します。
  - exostate()で外生状態変数を指定します。これはショックから影響を受ける変数です。

- observed ()で観測可能なコントロール変数を指定します。
- unobserved()は非観測な潜在コントロール変数を設定します。
- endostate()は内生状態変数で、ショックの影響は受けません。
- 各変数は一か所でだけ利用でき、オプションも一つだけ指定できます。
- 外生状態変数の個数と観測可能なコントロール変数の個数は等しいものとします。

- パラメータは大カッコで囲みます。
- パラメータは複数の式で利用できます。例えば、{beta}が何回も利用されていますが、これらは同一のものを指しています。
- F. は期待値を意味する演算子です。時系列シリーズの場合のリード項ではありませんので、注意してください。
- (4a)と(5a)式で期待値演算子E_tを利用しています。非線形 DSGE モデルの場合、期待値演算子は各推定式の先頭に入力します。

```
(1 = {beta}*(x/F.x)*(1/g)*(r/F.p))
```

これは次式に示す(4a)を表現しています。

$$E_t\left(1-\beta\frac{X_t}{X_{t+1}}\frac{1}{G_t}\frac{R_t}{\Pi_{t+1}}\right)=0$$

非線形 DSGE モデルのパラメータの推定と解釈

- 線形近似法を利用してモデルパラメータを推定します。連立式から線形な状態空間モデルを構築し、最尤法を用いて推定を行います。オプション nolog で反復ログを非表示にします。

```
dsgenl (1 = {beta}*(x/F.x)*(1/g)*(r/F.p))             ///
       (1/{phi} + (p-1) = {phi}*x + {beta}*(F.p-1))   ///
       ({beta}*r = p^(1/{beta})*u)                    ///
       (ln(F.u) = {rhou}*ln(u))                       ///
       (ln(F.g) = {rhog}*ln(g)),                      ///
       exostate(u g) observed(p r) unobserved(x) nolog
```

```
Solving at initial parameter vector ...
Checking identification ...

First-order DSGE model

Sample: 1954q3 - 2016q4                         Number of obs     =        250
Log likelihood = -768.09383
------------------------------------------------------------------------------
             |                 OIM
             |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
/structural  |
        beta |   .5112883   .0757912     6.75   0.000     .3627403    .6598363
         phi |   5.895207    1.65266     3.57   0.000     2.656052    9.134362
        rhou |   .6989192   .0449176    15.56   0.000     .6108823     .786956
        rhog |   .9556407   .0181345    52.70   0.000     .9200978    .9911836
-------------+----------------------------------------------------------------
     sd(e.u) |   2.317588   .2988029                      1.731945    2.903231
     sd(e.g) |   .6147347   .0973242                      .4239827    .8054867
------------------------------------------------------------------------------
```

- 推定結果の表に **beta** の推定値が出力されています。**beta** は 2 つの役割を果たしています。すなわち、一つは家計と企業の割引率であり、その逆数は利子率の推定式でインフレ率に対する中央銀行の応答を示すパラメータとなっています。
- 点推定値は 0.5 で、逆数は 2 となっています。

- 推定後の分析として、**beta** の逆数を **nlcom** コマンドで計算します。

```
nlcom 1/_b[beta]
```

```
       _nl_1:  1/_b[beta]

------------------------------------------------------------------------------
             |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
       _nl_1 |   1.955844   .2899259     6.75   0.000     1.387599    2.524088
------------------------------------------------------------------------------
```

一般的にこの値は 1.5 程度と言われています。推定値は 2 となっていますが、95%信頼区間に 1.5 が含まれています。

例題: 線形 DSGE モデル

DSGE モデルを書き換える

前述のモデル(4a)-(8a)は非線形モデルです。線形 DSGE モデルのコマンド **dsge** を利用する場合は線形変換する必要があります。(4a)-(8a)を線形変換した式を次に示します。

$$
\begin{aligned}
x_t &= E_t x_{t+1} - (r_t - E_t \pi_{t+1} - g_t) && (4b)\\
\pi_t &= \beta E_t \pi_{t+1} + \kappa x_t && (5b)\\
r_t &= \frac{1}{\beta}\pi_t + u_t && (6b)\\
u_{t+1} &= \rho_u u_t + \epsilon_{t+1} && (7b)\\
g_{t+1} &= \rho_g g_t + \xi_{t+1} && (8b)
\end{aligned}
$$

- 新しいパラメータκは(5a)式のパラメータϕの関数となっています。
- ここでは利子率の係数に-1 という制約を掛けます。

dsge コマンドでモデルを設定する

線形 DSGE モデルは(4b)-(8b)式によって定義されています。これをコマンドで示すと次のようになります。**///**は改行を示します。

```
. dsge  (p = {beta}*F.p + {kappa}*x)                ///
        (x = F.x - (r - F.p - g), unobserved)       ///
        (r = (1/{beta})*p + u)                      ///
        (F.u = {rhou}*u, state)                     ///
        (F.g = {rhog}*g, state)
```

- 非線形 DSGE モデルの **dsgenl** コマンドと同じく、カッコで式を囲みます。
- 変数の個数分だけ推定式を用意します。
- すべての変数が左辺に一度は表記される形にします。
- 右辺に記述する推定式は、ただ 1 つの内生変数に対応するものとします。
- 推定式の並び順は任意です。
- 変数の役割設定は **dsgenl** コマンドとは異なります。カッコを利用した推定式のオプションで、左辺の変数の役割が決まります。
  - ・ オプション **unobserved** で左辺の変数を非観測なコントロール変数とします。

- オプション state は状態変数として設定します。
- p と r は観測可能なコントロール変数p_tとr_tです。この 2 つの推定式ではオプションは利用しません。
- 産出ギャップx_tは非観測のコントロール変数ですからオプションの unobserved を利用します。
- u と g は状態変数のu_tとg_tです。ここではオプションの state を利用します。
- 状態変数は現時点では固定されています。1 時点先の値を F.を使ってモデリングします。

- ϵ_tとξ_tは変数のショックを示し、状態方程式で利用します。ショックの個数と観測可能なコントロール変数の個数は一致させます。例えば、(7b)式を次のように記述したとします。

```
. dsge ... (F.u = {rhou}*u, state) ...
```

  - モデルで状態変数を確定的な変数として扱うのであれば、このようにショックは不要です。
  - 例えば、資本蓄積はこのように確定的な変数として表現する場合があります。
  - ショックを持たない状態変数を推定式に含む場合は、noshock オプションを利用します。

- dsgenl コマンドと同じく、F.演算子は状態およびコントロール変数に対して利用可能で、これは変数の期待値を意味します。
  - 例えばx_{t+1}を表現する場合は F.x とします。ただし、時系列x_tの t+1 時点における期待値ではなく、モデル側から見たときの期待値を意味します。
  - 次のコマンドの場合について考えてみましょう。

```
. dsge (p = {beta}*F.p + {kappa}*x) ...
```

    これは次のように解釈されます(4b と同じです)。

$$E_t(p_t - \beta p_{t+1} - \kappa x_t) = 0$$

  - 線形 DSGE モデルの場合、期待値演算子の対象が変数、推定式、またはシステム全体なのかはあまり、問題ではありません。
  - ただし、推定するパラメータには大カッコを付けることを忘れないでください。

線形 DSGE モデルにおけるパラメータ推定と解釈

金融経済学におけるモデル(4b)-(8b)を推定します。

- (4b)は産出ギャップのオイラー方程式です。
- (5b)はニューケイジアン フィリプス曲線でκはフィィリップス曲線の傾きです。ニューケイジアンモデルにおいて価格は産出量に依存し、κは依存性を示します。
- (6a)はテイラールールです。

```
dsge (p = {beta}*F.p + {kappa}*x)           ///
      (x = F.x - (r - F.p - g), unobserved) ///
      (r = (1/{beta})*p + u)                ///
      (F.u = {rhou}*u, state)               ///
      (F.g = {rhog}*g, state)
```

```
(setting technique to bfgs)
Iteration 0:   log likelihood = -13931.564
Iteration 1:   log likelihood = -1301.5118  (backed up)
Iteration 2:   log likelihood = -1039.6984  (backed up)
Iteration 3:   log likelihood = -905.70867  (backed up)
Iteration 4:   log likelihood = -842.76867  (backed up)
(switching technique to nr)
Iteration 5:   log likelihood = -812.04209  (backed up)
Iteration 6:   log likelihood = -786.76609
Iteration 7:   log likelihood = -777.19779
Iteration 8:   log likelihood =  -768.8383
Iteration 9:   log likelihood =  -768.1368
Iteration 10:  log likelihood = -768.09519
Iteration 11:  log likelihood = -768.09383
Iteration 12:  log likelihood = -768.09383

DSGE model

Sample: 1954q3 - 2016q4                         Number of obs     =        250
Log likelihood = -768.09383
------------------------------------------------------------------------------
             |                 OIM
             |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
/structural  |
        beta |   .5112881    .075791     6.75   0.000     .3627404    .6598359
       kappa |   .1696296   .0475492     3.57   0.000      .076435    .2628243
        rhou |   .6989189   .0449192    15.56   0.000     .6108789    .7869588
        rhog |   .9556407   .0181342    52.70   0.000     .9200983    .9911831
-------------+----------------------------------------------------------------
     sd(e.u) |   2.317589   .2988024                      1.731947     2.90323
     sd(e.g) |   .6147348   .0973279                      .4239757    .8054939
------------------------------------------------------------------------------
```

- パラメータ kappa はフィリップス曲線の傾きです。理論的には正値となりますが、実際に推定値は正となっています。
- パラメータ beta は利子率推定式におけるインフレ率の係数の逆数です。ここでは$1/\beta$として計算されますが、これはインフレ対する中央銀行の反応の程度と理解できます。

- 推定後の分析として、beta の逆数を nlcom コマンドで計算します。

```
nlcom 1/_b[beta]
```

```
      _nl_1:  1/_b[beta]

------------------------------------------------------------------------------
             |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
       _nl_1 |   1.955844   .2899255     6.75   0.000     1.387601    2.524088
------------------------------------------------------------------------------
```

- 一般的に$1/\beta$は 1.5 と言われています。この例題の推定値は約 2 となりました。

診断

　dsgenl コマンド、dsge コマンドによるパラメータ推定後に利用可能な診断機能を紹介します。ここでは dsge コマンドを実行した例題をそのまま利用します。ここで利用したモデルの場合、dsgenl コマンドを利用したとしても、結果は同じものになります。

政策行列および遷移行列

- 状態空間におけるパラメータ行列を利用して、状態変数がコントロール変数に与える影響を設定します。ここでいうコントロール変数とは政策行列を指します。
- 政策行列パラメータは、状態変数に対する 1 単位のショックの効果と理解できます。
- estat policy コマンドを実行して、政策行列を表示します。

```
estat policy
```

Policy matrix

	Coef.	Delta-method Std. Err.	z	P>|z|	[95% Conf.	Interval]
p						
u	-.4170859	.0389324	-10.71	0.000	-.4933919	-.3407799
g	.881884	.2330573	3.78	0.000	.4251001	1.338668
x						
u	-1.580153	.3926336	-4.02	0.000	-2.3497	-.8106049
g	2.658667	.9045286	2.94	0.003	.885823	4.43151
r						
u	.1842449	.056798	3.24	0.001	.072923	.2955669
g	1.724828	.2210259	7.80	0.000	1.291625	2.158031

- 推定結果は推定式単位で表示されます。
- 先頭のブロックはインフレ率 p に対する政策推定式で、状態変数の関数として表示されています。
  - 状態 u に対するショックはインフレ率を 0.417 だけ減少させます。逆に g に対するショックは 0.88 だけインフレ率を押し上げます。

- 状態変数のダイナミックな関係を定義するパラメータ行列は状態遷移行列と呼ばれています。
- estat transition コマンドを実行して、状態遷移行列を表示します。

```
estat transition
```

```
Transition matrix of state variables

------------------------------------------------------------------------------
             |                 Delta-method
             |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
F.u          |
           u |   .6989189   .0449192    15.56   0.000     .6108789    .7869588
           g |   1.11e-16          .        .       .            .           .
-------------+----------------------------------------------------------------
F.g          |
           u |          0  (omitted)
           g |   .9556407   .0181342    52.70   0.000     .9200983    .9911831
------------------------------------------------------------------------------
Note: Standard errors reported as missing for constrained transition matrix
      values.
```

- 状態変数はどちらも自己回帰過程を使ってモデル化されます。
- 状態遷移行列の片方はゼロ、または、ゼロからの差分となり、関連する統計量は表示しません。

インパルス応答

- 状態変数へのショックによるコンロール変数や状態変数の反応をトレースすることが可能です。このパスのことをインパルス応答(IRF: impulse-response function)と呼びます。
- IRF の情報は .irf ファイルに格納され、**irf graph** や **irf table** コマンドで画面表示可能です。
- 次のコマンドは IRF ファイルを作成し、アクティブに設定します。

```
irf set nkirf.irf
```

```
(file nkirf.irf created)
(file nkirf.irf now active)
```

- **dsge** コマンドでモデルを推定した後で、IRF のすべての情報を取得する場合は **irf create** コマンドを利用します。
- この例題の場合、**irf create** コマンドで **e.u** と **e.g** のショックを使って **r, g, u** に対するインパルス応答の結果を保存します。

●　結果は nkirf.irf に保存されます。

```
irf create model1
```

```
(file nkirf.irf updated)
```

●　irf graph コマンドでインパルス応答のグラフを作成します。
●　応答を表示する場合は、次のようにします。

```
irf graph irf, impulse(u) response(x p r u) byopts(yrescale)
```



●　利子率における状態変数 u は、インフレ率と利子率間のフィードバック効果以外の、利子率の上昇要因の動きをモデル化しています。
●　u に対するショックは利子率を増加させ、また、IRF により生産ギャップに対する出力を減少させることが分かります。

予測

forecast の一連のコマンドセットはフィットしたモデルのダイナミック予測を計算します。

- （事前準備その 1）dsge の推定結果を保存します。

```
estimates store dsge_est
```

- （事前準備その 2）データの期間を 3 年から 12 四半期に変更します。

```
tsappend, add(12)
```

- 予測は次の 3 ステップで実行します。
  1. forecast create で新たな予測モデルを初期化します。

```
forecast create dsgemodel
```

```
Forecast model dsgemodel started.
```

2. dsge コマンドによる推定結果を forecast estimates コマンドで予測モデルに追加します。このコマンドは dsge_est をモデル dsgemodel に追加します。

```
forecast estimates dsge_est
```

```
Added estimation results from dsge.
Forecast model dsgemodel now contains 2 endogenous variables.
```

3. 2017 年第一四半期を始期とするダイナミック予測を forecast solve コマンドで実行します。オプション begin(tq(2017q1))を使用します。

```
forecast solve, prefix(d1_) begin(tq(2017q1))
```

```
Computing dynamic forecasts for model dsgemodel.
------------------------------------------------

Starting period:  2017q1
Ending period:    2020q1
Forecast prefix:  d1_

2017q1:  .............
2017q2:  .............
2017q3:  .............
2017q4:  .............
2018q1:  .............
2018q2:  .............
2018q3:  .............
2018q4:  .............
2019q1:  .............
2019q2:  .............
2019q3:  .............
2019q4:  .............
2020q1:  .............

Forecast 2 variables spanning 13 periods.
-----------------------------------------
```

2017 年第一四半期から始まるダイナミック予測が計算されます。この予測はアウトオブサンプルの予測となります。

- tsline コマンドを使ってインフレ率 d1_p の予測値を求めます。

```
tsline d1_p if tin(2010q1, 2021q1), tline(2017q1)
```



長期的には標本平均にゆっくり回帰する様子が描かれます。

- 観測値が入手可能な in-sample の予測も可能です。`begin(tq(2014q1))`オプションを使って 2014 年第一四半期を始期とするダイナミック予測を実行します。
- 2014-2016 年の間で予測値と実現値を比較します。

```
forecast solve, prefix(d2_) begin(tq(2014q1))
```

```
Computing dynamic forecasts for model dsgemodel.
------------------------------------------------

Starting period:  2014q1
Ending period:    2020q1
Forecast prefix:  d2_

2014q1:  .............
2014q2:  .............
2014q3:  .............
2014q4:  .............
2015q1:  .............
2015q2:  .............
2015q3:  .............
2015q4:  ............
2016q1:  ............
2016q2:  ............
2016q3:  ............
2016q4:  ............
2017q1:  ............
2017q2:  ............
2017q3:  ............
2017q4:  ............
2018q1:  ............
2018q2:  ............
2018q3:  ............
2018q4:  ............
2019q1:  ............
2019q2:  ............
2019q3:  ............
2019q4:  ............
2020q1:  ............

Forecast 2 variables spanning 25 periods.
-----------------------------------------
```

- 実現値と予測値をプロットします。

```
tsline p d2_p if tin(2010q1, 2021q1), tline(2014q1)
```



- 予測値は 2014-2016 年の上昇トレンドを捉えています。
- しかし、上昇トレンドの周囲の変動はモデル化できていません。